Maple-assisted proof of formula for A251366

Robert Israel

19 March 2018

Of the $3^4 = 81$ possible configurations for a 2×2 subblock (where $[a, b, c, d]$ encodes $\begin{bmatrix} a & b \\ c & d \end{bmatrix}$, all but two sum to 1 to 7.

```
> Blockconfigs:= [seq(convert(x,base,3)[1..4],x=3^5+1..3^5+79)];
```

Blockconfigs := [[1, 0, 0, 0], [2, 0, 0, 0], [0, 1, 0, 0], [1, 1, 0, 0], [2, 1, 0, 0], [0, 2, 0, 0], [1, 2, 0, 0], [2, 2, 0, 0], [0, 0, 1, 0], [1, 0, 1, 0], [2, 0, 1, 0], [0, 1, 1, 0], [1, 1, 1, 0], [2, 1, 1, 0], [0, 2, 1, 0], [1, 2, 1, 0], [2, 2, 1, 0], [0, 0, 2, 0], [1, 0, 2, 0], [2, 0, 2, 0], [0, 1, 2, 0], [1, 1, 2, 0], [2, 1, 2, 0], [0, 2, 2, 0], [1, 2, 2, 0], [2, 2, 2, 0], [0, 0, 0, 1], [1, 0, 0, 1], [2, 0, 0, 1], [0, 1, 0, 1], [1, 1, 0, 1], [2, 1, 0, 1], [0, 2, 0, 1], [1, 2, 0, 1], [2, 2, 0, 1], [0, 0, 1, 1], [1, 0, 1, 1], [2, 0, 1, 1], [0, 1, 1, 1], [1, 1, 1, 1], [2, 1, 1, 1], [0, 2, 1, 1], [1, 2, 1, 1], [2, 2, 1, 1], [0, 0, 2, 1], [1, 0, 2, 1], [2, 0, 2, 1], [0, 1, 2, 1], [1, 1, 2, 1], [2, 1, 2, 1], [0, 2, 2, 1], [1, 2, 2, 1], [2, 2, 2, 1], [0, 0, 0, 2], [1, 0, 0, 2], [2, 0, 0, 2], [0, 1, 0, 2], [1, 1, 0, 2], [2, 1, 0, 2], [0, 2, 0, 2], [1, 2, 0, 2], [2, 2, 0, 2], [0, 0, 1, 2], [1, 0, 1, 2], [2, 0, 1, 2], [0, 1, 1, 2], [1, 1, 1, 2], [2, 1, 1, 2], [0, 2, 1, 2], [1, 2, 1, 2], [2, 2, 1, 2], [0, 0, 2, 2], [1, 0, 2, 2], [2, 0, 2, 2], [0, 1, 2, 2], [1, 1, 2, 2], [2, 1, 2, 2], [0, 2, 2, 2], [1, 2, 2, 2]] **(1)**

Consider the 79×79 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×2 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j), and 0 otherwise. The following Maple code computes it.

```
> T:= Matrix(79,79,(i,j) -> `if`(Blockconfigs[i][1..2]=Blockconfigs
  [j][3..4],1,0));
```

$$T := \begin{bmatrix} \textit{79 x 79 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \quad \textbf{(2)}$$

Thus for $n \geq 1$, $a(n) = u\, T^{n-1}\, v$ where u and v are 79-dimensional row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](79, 1):
  v:= Vector(79,1):
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^(n-1) . v,n=1..11);
```

79, 695, 6113, 53769, 472943, 4159927, 36590017, 321839625, 2830847119, 24899654327, 219013164449 **(3)**

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= LinearAlgebra:-MinimalPolynomial(T, t);
```

$$P := t^5 - 7\,t^4 - 15\,t^3 - 7\,t^2 \quad \textbf{(4)}$$

The empirical formula corresponds to a factor of this.

```
> factor(P);
```

$$t^2\,(t+1)\,(t^2 - 8\,t - 7) \quad \textbf{(5)}$$

If $r(n) = a(n+3) - 8\,a(n+2) - 7\,a(n+1) = u\cdot\left(T^{n+2} - 8\,T^{n+1} - 7\,T^n\right)\,v$, then $r(n) + r(n+1) = u\,T^n(T+I)\left(T^2 - 8\,T - 7\,I\right)\,v = 0$ for $n \geq 2$.

Given that $r(0) = r(1) = r(2) = 0$, we conclude that all $r(n) = 0$, i.e. the empirical formula is true.